HOMEWORK 1

AN ANALYSIS OF AN INCREASE OF ONE FACTOR'S ENDOWMENT AND TAXATION ON ALL FACTORS

by

Fikrah Elhifzi Harahap (437961)

Ankit Pandey (437949)

Nurseto Dwi Nugroho (448052)

Tasks:

1. Run a simulation of an increase of one factor's endowment
2. Run a simulation of a uniform tax on all sectors/factors
3. How are the results different from when you introduced the tax on only one good and factor When taxes are distortionary?

Solutions:

The full code in GAMS

```
$ONTEXT
        X    Y    L    K    W    CONS      Total
X                          100                100
Y                          100                100
W                                    200      200
L         40   60                              100
K         60   40                              100
CONS                100  100                   200
Total     100       100  100 100  200  200
$OFFTEXT
SETS
    SEC /X,Y/
    FAC /K,L/
    ;
PARAMETERS
    BETA(FAC,SEC)
    ALPHA(SEC)
    XS0(SEC)
    W0
    OMEGA(FAC)
    USE0(FAC,SEC)
    TAX(FAC,SEC)
    ;
* USE0 is the factor use by sector
USE0("K","X") = 60;
USE0("L","X") = 40;
USE0("K","Y") = 40;
USE0("L","Y") = 60;

* Benchmark outputs (from SAM)
XS0(SEC) = 100;


* Assigns numbers from SAM to parameters:
```

```
* Parameters of the Cobb Douglas production function
BETA(FAC,SEC) = USE0(FAC,SEC) / XS0(SEC);

* Parameters of the Cobb Douglas utility function
ALPHA(SEC) = 0.5;

* Benchmark utility
W0 = 200;

* Endowments of factors
OMEGA(FAC) = 100;

* Important: we have to introduce the tax in all sectors/factors but we set it to zero.
TAX(FAC,SEC) = 0;

* The Cobb Douglas scaling factor
* You need that in order for the output and total consumption aggregate
* to be exactly as much as you want.

PARAMETERS
A
B(SEC);

* For consumption:
A   = W0 / (PROD(SEC, XS0(SEC)**ALPHA(SEC)));
* For Production
B(SEC) =  XS0(SEC) / ( PROD(FAC, USE0(FAC,SEC) ** BETA(FAC,SEC)));


VARIABLES
    XS(SEC)             Output
    P(SEC)              Prices of final goods
    W                   Welfare index
    PW                  Consumer price index
    PFAC(FAC)           Factor wages
    INC                 Consumer income
    TRICK
;

EQUATIONS
    PRF_XS(SEC)      Zero profits on goods production production
    PRF_W           Determination of the consumer price index

    MKT_XS(SEC)      Market clearing for goods
    MKT_W           Demand for aggregate consumption

    MKT_FAC(FAC)     Market clearing for factors

    I_INC           Income balance
    TRCK
;
*(PFAC(FAC)*(1+TAX(FAC,SEC)))

PRF_XS(SEC)..
    PROD(FAC, ((PFAC(FAC)*(1+TAX(FAC,SEC)))/BETA(FAC,SEC))**BETA(FAC,SEC)) / B(SEC)
=E= P(SEC);

PRF_W..
    PROD(SEC, (P(SEC) / ALPHA(SEC))**ALPHA(SEC))  / A =E= PW;

MKT_XS(SEC)..
    XS(SEC) =E= ALPHA(SEC) * W * PW / P(SEC);

MKT_W..
    W =E= INC / PW  ;

MKT_FAC(FAC)..    OMEGA(FAC) =E= SUM(SEC, BETA(FAC,SEC) * P(SEC) * XS(SEC)
                        / (PFAC(FAC)* (1 + TAX(FAC,SEC)))) ;
```

```
I_INC..              INC =E= SUM(FAC, PFAC(FAC) * OMEGA(FAC))
                  + SUM(SEC, SUM(FAC, TAX(FAC,SEC) * PFAC(FAC)
                  * BETA(FAC,SEC) * P(SEC) * XS(SEC)
                       / (PFAC(FAC)* (1 + TAX(FAC,SEC))))) ;


TRCK..  TRICK=E=1;

MODEL SIMPLE
     /PRF_XS
    PRF_W
    MKT_XS
    MKT_W
    MKT_FAC
    I_INC
    TRCK
/
;

* Provide starting values for the solver

     XS.L(SEC)= XS0(SEC);
     W.L= W0;
     P.L(SEC)= 1;
     PW.L= 1;
     PFAC.L(FAC)= 1;
     INC.L= 200;

* Provide lower bounds for the solver
     XS.LO(SEC)= 0.00001;
     W.LO= 0.00001;
     P.LO(SEC)= 0.00001;
     PW.LO= 0.00001;
     PFAC.LO(FAC)= 0.00001;
     INC.LO= 0.00001;

* Provide a numeraire:
PFAC.FX("K")=1;

*simple.iterlim=0;

option nlp = pathnlp;
SOLVE  SIMPLE USING NLP maximizing TRICK;

* uniform tax
TAX(FAC,SEC)=0.6;
SOLVE SIMPLE USING NLP maximizing TRICK;
*factor endowment
TAX(FAC,SEC) = 0;
OMEGA("K") = 1.2* OMEGA("K");
SOLVE SIMPLE USING NLP maximizing TRICK;

* OMEGA(FAC) factor demand
$onExternalOutput

Variable INC 'INC';
Parameter OMEGA(FAC) 'OmegaFactor'
$offExternalOutput
```

1. Simulation of an Increase of one factor's endowment

From the previous model, we have chosen that the Capital (K) will be the factor endowment to be increased. By the following table, you can see the changes on every variables when we applied different amount of capital.

Table 1. Input-output changes in all variables towards one factor's endowment (capital)

	OMEGA								
	TAX=0	100	120	140	160	180	200	220	240
Variables									
Consumer Income	INC	200	240	280	320	360	400	440	480
Consumer Price Index	PW	1.0	1.09	1.18	1.26	1.34	1.41	1.48	1.55
Welfare Index	W	200	219.08	236.64	252.98	268.32	282.8	296.6	309.8
Price of final goods X	PX	1.0	1.07	1.14	1.20	1.27	1.32	1.37	1.42
Price of final goods Y	PY	1.0	1.12	1.22	1.32	1.42	1.52	1.60	1.69
Output price of good X	XS(SEC)	100	111.56	122.37	132.57	142.28	151.6	160.5	169.1
Output price of good Y		100	107.56	114.40	120.68	126.51	131.9	137.1	141.9
Factor Wages (K)	PFAC(FAC)	1	1	1	1	1	1	1	1
Factor Wages (L)		1	1.2	1.4	1.6	1.8	2	2.2	2.4

As we can see from table 1, the increase of capital (K) is 20% which affects the change of the other variables too except the factor wages for the capital. This is in line with **Rybczynski Theorem** which says that "At a given commodity price, if the endowment of some resource increases, the industry that uses that resource relatively intensively will increase its output, while the other industry reduces its output".

Source:

1. Testing Rybczynski Theorem: An Evidence from The Selected European Transition Countries, Mediterranean Journal of Social Sciences (2013)

Look at the table 1 with yellow color. We can see that the variables are changing since the capital is increased by 20%. By the following Graph 1, it is clear that the variables are increased as the capital is increased, except the factor wages of capital (K).

Graph 1. Input-output Changes in All Variables Towards 20% of Increase in Capital



To have a more detail look, the graphs which shows the relation between omega and each variable are presented below.

Graph 2. Graph of Omega vs Consumer Income



Graph 3. Graph of Omega vs Consumer Price Index



Graph 4. Graph of Omega vs Welfare Index



Graph 5. Graph of Omega vs Price of Final Good X



Graph 6. Graph of Omega vs Price of Final Good Y



Graph 7. Graph of Omega vs Output Price of Good X



Graph 8. Graph of Omega vs Price of Final Good Y



According to these results from the graphs, we can conclude that that the variables are increasing as the one factor's endowment increases which means that the capital, as the one factor's endowment affects the consumer income, consumer price index, welfare index, price of final good x and y, output price of good x and y, and factor wages/endowment L. But, interestingly, the factor wages of capital stay the same. This shows that the increase of the capital doesn't affect the factor wages of capital.

2. Simulation of a uniform tax on all sectors/factors

Under the previous model, we will now apply a tax to both sectors and factors. After applying the tax on the formula needed in the code, the following table is gained.

Table 2. Input-output changes in all variables towards an application of uniform taxation under 20% increase in capital as one factor's endowment

Variable	**Initial**	**Uniform TAX(SEC/FAC)**
Consumer Income	200	240
Consumer Price Index	1	1.2
Welfare Index	200	200

Price of final goods X	1	1.2
Price of final goods Y	1	1.2
Output price of good X	100	100
Output price of good Y	100	100
Factor Wages (K)	1	1
Factor Wages (L)	1	1

It is interesting that when the final price of goods X and Y have increased, the output price of X and Y remain the same. For endowment's factor K and L, there is no change, because tax has the same impact for both factors, also the welfare index stays the same. But on the other side, the higher the tax, the bigger the increase of consumer income and price index.

Graph 9. Input-output Changes in All Variables Towards 20% of Increase in Capital



To summarize, the uniform tax will affect the income and the final price that will be given to the end customer. Anyway, the increment of income doesn't mean that consumers are in better condition, but statistically, welfare or the ability to afford goods X and Y stay the same whether the income is the same or it increases. As additional analysis for the behaviour of the prices of final goods, consumer price index and income, more experiments have been done. Particularly for several different values of taxes the model has been run.

Table 3. Input-output changes in all variables towards applied taxation and increase of capital of 20%

	TAX & Increase of Capital (K) of 20%								
	TAX	0	0.2	0.4	0.6	0.8	1	1.2	1.4
Variables									
Consumer Income	INC	200	240	280	320	360	400	440	480

Consumer Price Index	PW	1	1.2	1.4	1.6	1.8	2	2.2	2.4
Welfare Index	W	200	200	200	200	200	200	200	200
Price of final goods X	PX	1	1.2	1.4	1.6	1.8	2	2.2	2.4
Price of final goods Y	PY	1	1.2	1.4	1.6	1.8	2	2.2	2.4
Output price of good X	XS(SEC	100	100	100	100	100	100	100	100
Output price of good Y		100	100	100	100	100	100	100	100
Factor Wages (K)	PFAC(FAC)	1	1	1	1	1	1	1	1
Factor Wages (L)		1	1	1	1	1	1	1	1

Graph 10. Graph of Consumer Income vs Tax



Graph 11. Graph of Consumer Price Index vs Tax



Graph 12. Graph of Consumer Income vs Tax

Based on the graphs 10 to 12, the behavior of these variables is perfectly linear when the value of the tax is changing. The increase of these variables is proportional to the increase of the taxes.

Sources: 1. Tax – Wikipedia

3. How are the results different from when you introduced the tax on only one good and factor When taxes are distortionary?

The increase will remain the same, 20%. In here, an uniform tax on capital (K) is introduced in sector X → "K", "X".

Table 4. Input-output changes in all variables towards an application of uniform taxation on only sector Y under 20% increase in capital

Scalar	**Initial**	**Uniform TAX on capital in sector X**
Consumer Income	200	222,22

Consumer Price Index	1	1,11
Welfare Index	200	199,60
Price of Final a good X	1	1,16
Price of Final a good Y	1	1,07
Output Price of a good X	100	95,49
Output Price of a good Y	100	104,30
Factor Wages (K)	1	1
Factor Wages (L)	1	1

While some variables such as welfare index, output price and one of comparable goods are inversely proportional, the income, price index and final price are produced positively by the uniform tax on capital in sector X. Simply, the tax on X means the price will become higher. It makes the consumers choose substitutions product and when it happens it affects the output price

We can assume that The final price is increased while the output price is decreased. It happens to get the equilibrium. So, when the tax is given, they will 'play' to get the customer.

Graph 12. Input-output Changes in All Variables Towards Application of a Uniform Tax in Sector X



For the second case, uniform Tax on capital in sector Y is introduced, vice versa from the first case.

Table 5. Input-output changes in all variables towards an application of uniform taxation on only sector Y under 20% increase in capital

Scalar	**Initial**	**Uniform TAX on capital in sector Y**
Consumer Income	200	214,29
Consumer Price Index	1	1,07
Welfare Index	200	199,61
Price of Final a good X	1	1,03
Price of Final a good Y	1	1,12
Output Price of a good X	100	95,49
Output Price of a good Y	100	104,30
Factor Wages (K)	1	1
Factor Wages (L)	1	1

For the second table, it is interesting because the price is a bit different. Now both final price and the output price are increased compared to initial state for Y. We assume that fact that the product Y awareness is strong, so any changes won't affect them too much.

Graph 12. Input-output Changes in All Variables Towards Application of a Uniform Tax in Sector Y



For this case, uniform tax is given on labor in sector X → "L", "X" and sector Y → "L", "Y"

Table 6. Input-output changes in all variables towards an application of uniform taxation on sector X and Y under 20% increase in capital

Scalar	**Initial**	**Uniform TAX on labor in sector X**	**Uniform TAX on labor in sector Y**
Consumer Income	200	200,00	214,29
Consumer Price Index	1	1	1,07
Welfare Index	200	199,61	199,61
Price of Final a good X	1	1,05	1,03
Price of Final a good Y	1	0,96	1,12
Output Price of a good X	100	95,57	95,57

Output Price of a good Y	100	104,23	104,23
Factor Wages (K)	1	1	1
Factor Wages (L)	1	1	1

Based on this table, Labor affects the product. It makes the price of final goods change. On the other hand, when tax applies in labor for sector X in the consumer income and price index, the numbers have not shifted.